

Solving the Shallow Water Equations using Finite Volumes and Lax-Friedrichs

Fundamentals of Wave Simulation Seminar

Nathan W Brei

Fakultät für Informatik

Technische Universität München

Email: brei@in.tum.de

Abstract—Lax-Friedrichs and Local Lax-Friedrichs are two numerical methods for solving the shallow water equations. This paper introduces both and provides a concise implementation in Julia. The Gaussian perturbation and breaking dam model problems are solved for each method over a range of parameters. The cause and effects of artificial viscosity are explored analytically and experimentally.

Keywords—Shallow water equations, Finite volumes, Lax-Friedrichs method, Rusanov, artificial viscosity, Julia

I. INTRODUCTION

This paper introduces two finite volume methods for solving the shallow water equations, and applies them to the Gaussian perturbation and breaking dam problems. The first three sections respectively introduce the shallow water equations, the general finite volume method, and the CFL condition. Section 5 derives the Lax-Friedrichs method, and Section 6 demonstrates that it captures the desired dynamics but exhibits undesired viscous behavior. Section 7 shows that this viscosity arises from the method's numerics. Section 8 introduces the Local Lax-Friedrichs (Rusanov) method, and Section 9 shows that the viscous behavior is largely mitigated. Section 10 discusses a concise Julia implementation, which is provided in the Appendix.

II. SHALLOW WATER EQUATIONS

The shallow water equations are a system of hyperbolic partial differential equations which are effective at describing the behavior of tsunamis and overland flows. [1] They are derived from the general Navier-Stokes equations by making the following assumptions:

- In the 1-D case, the flow is in a channel of unit width.
- Horizontal velocity $u(x)$ across any cross section x is constant.
- Vertical velocity $v(x)$ is small (but not zero) and falls out of the equations.
- Pressure is dominated by hydrostatics: $p = \frac{1}{2}\rho gh^2$

The 1D equations are as follows:

$$\begin{bmatrix} h \\ hu \end{bmatrix}_t + \begin{bmatrix} hu \\ hu^2 + \frac{1}{2}gh^2 \end{bmatrix}_x = 0$$

The shallow water equations may be defined for a state vector $q := (h, hu)$ in an abstract way, permitting a decoupling between the underlying equation and the numerical method.

Because the shallow-water equations match the pattern of a general conservation law, $q_t(x, t) + f(q(x, t))_x = 0$, the dynamics are captured by the pointwise flux function f , where

$$f : (q_1, q_2) \mapsto \left(q_2, \frac{q_2^2}{q_1} + \frac{gq_1^2}{2} \right) \quad (1)$$

The shallow water equations enter the method in only one other place, the max wave speed function. The system's wavespeeds are given by the eigenvalues of the flux Jacobian,

$$f'(q) = \begin{bmatrix} 0 & 1 \\ -u^2 + gh & 2u \end{bmatrix}$$

and the absolute value of the dominating eigenvalue,

$$\lambda_{max} = \max |u \pm \sqrt{gh}| \quad (2)$$

will be shown to play an important role in ensuring numerical stability.

III. FINITE VOLUMES

Finite volumes are a family of numerical methods which are well suited for computing an approximate solution to hyperbolic systems such as the shallow water equations. [2] They have two main advantages over methods such as finite elements. Firstly, they automatically conserve physical quantities q . Secondly, they can capture discontinuities, such as shocks, which is considerably simpler than explicitly tracking them. One downside is that the profile of the shock will be smeared by the spatial discretization.

The starting point is to recast the conservation equations in integral form:

$$\frac{d}{dt} \int_{C_i} q(x, t) dx = f(q(x_{i-1/2}, t)) - f(q(x_{i+1/2}, t))$$

Since these equations hold for an arbitrary control volume, we partition the spatial domain into cells, rather than points. For simplicity these cells are assumed to be regular, although the equations hold just as well over an unstructured mesh.

$$C_i := (x_{i-1/2}, x_{i+1/2})$$

After integrating with respect to time, each term takes on a physical significance directly relevant to our model:

$$\int_{C_i} q(x, t_{n+1}) dx - \int_{C_i} q(x, t_n) dx = \int_{t_n}^{t_{n+1}} f(q(x_{i-1/2}, t)) dt - \int_{t_n}^{t_{n+1}} f(q(x_{i+1/2}, t)) dt$$

- Let $Q_i^n \approx \frac{1}{\Delta x} \int_{C_i} q(x, t_n) dx$ be the approximate average q in cell i at time n .
- Let $F_{i+1/2}^n \approx \frac{1}{\Delta t} \int_{t_n}^{t_{n+1}} f(q(x_{i+1/2}, t)) dt$ be the approximate average flux f from cell i into $i+1$.

Rearranging leads to a general update scheme:

$$Q_i^{n+1} = Q_i^n - \frac{\Delta t}{\Delta x} (F_{i+1/2}^n - F_{i-1/2}^n) \quad (3)$$

When the cellwise flux function can be calculated using only the state in the adjacent cells, the method is fully discrete:

$$F_{i+1/2}^n = \mathcal{F}(Q_i^n, Q_{i+1}^n)$$

Taking a step back, it is clear that these methods also fall under the umbrella of finite differences. However, they have the additional property of being conservative: Even if the computed fluxes are rather inaccurate, the sum of fluxes over the spatial domain will vary only according to boundary conditions.

The desired properties of finite volumes hold for any choice of $F_{i+1/2}^n$, as long as the result is consistent and stable. Similarly, they hold for any choice of hyperbolic PDE, which is abstracted away as a state space Q_i , a pointwise flux $f(Q_i)$ and max wavespeed $\lambda_{max}(Q_i)$. This natural decoupling of the mathematical machinery can be directly exploited in the implementation by parameterizing a generic finite volume skeleton with different strategies.

IV. NUMERICAL STABILITY

In order to ensure numerical stability, a necessary (but not sufficient) condition that must be met is the *CFL condition*. For a hyperbolic system, information propagates through the spatial domain at finite speeds. Since the system evolves by exchanging fluxes between neighboring cells, it is intuitively necessary that the timestep size be small enough that the fluxes don't travel further than one cell per timestep. More precisely, the *numerical domain of dependence* (the cells whose old values affect each cell's new value) must contain the *physical domain of dependence*, the spatial region from which waves propagated to a given point, over that same time interval. This is expressed via the dimensionless Courant number, which relates cell size, timestep size, and wave speed. For a three-point stencil, the Courant number is constrained to:

$$\nu := \left| \frac{\bar{u} \Delta t}{\Delta x} \right| = \frac{\Delta t}{\Delta x} |\lambda_{max}| \leq 1 \quad (4)$$

Because this code uses the same timestep size for all cells, the maximum wave speed must be calculated across the entire spatial domain for each timestep iteration. The maximum wavespeed calculated from the flux Jacobian is exact only in

the small-perturbation case, so an additional correction factor, $\mu \in [0, 1]$, scales the timestep down. Thus the experimenter has two knobs for balancing speed, accuracy, and stability: the spatial discretization Δx and the timestep correction μ .

V. THE LAX-FRIEDRICHS METHOD

Different finite-volume methods may be derived from a common skeleton by providing a choice of \mathcal{F} . The simplest choice is merely the average of the fluxes at the center of either neighboring cell.

$$F_{i+1/2} = \frac{1}{2} [f(Q_i) + f(Q_{i+1})]$$

$$Q_i^{n+1} = Q_i^n - \frac{\Delta t}{2\Delta x} [f(Q_{i+1}^n) - f(Q_{i-1}^n)]$$

This yields a forward-time, centered-space finite difference scheme. Unfortunately a von Neumann analysis shows that it is always unstable. [3] The Lax-Friedrichs method simply modifies this scheme by changing the time difference to use a value of Q_i^n averaged from the neighboring cells:

$$Q_i^n \approx \frac{1}{2} (Q_{i-1}^n + Q_{i+1}^n)$$

The resulting update scheme becomes

$$Q_i^{n+1} = \frac{1}{2} [Q_{i-1}^n + Q_{i+1}^n] - \frac{\Delta t}{2\Delta x} [f(Q_{i+1}^n) - f(Q_{i-1}^n)]$$

This may be rearranged to yield a cell-wise flux function:

$$F_{i+1/2} = \frac{1}{2} [f(Q_{i+1}) + f(Q_i) - \frac{\Delta x}{\Delta t} (Q_{i+1} - Q_i)] \quad (5)$$

The parameter $a := \frac{\Delta x}{\Delta t}$ has multiple interpretations. Firstly, it was derived as the μ -scaled max wavespeed over the spatial domain. Secondly, it acts as a penalty term: the larger the discontinuity in $Q(x)$, the lower the cellwise flux $F(x)$. The following sections will explore its unexpected third interpretation, as a viscosity coefficient.

VI. LAX-FRIEDRICHS RESULTS

The Lax-Friedrichs method was run against two model problems: a Gaussian perturbation with reflecting boundaries, and a breaking dam with outflow boundaries. The independent behavior of the knobs n_{cells} and μ was tested in three experiments. The results are depicted in Figures 1, 2, and 3.

The experiment depicted in Figure 1 is rather intuitive. The Gaussian perturbation propagates symmetrically across the spatial domain. As the spatial discretization is refined, a shock wave gradually emerges from each (previously symmetric) ripple. It should be noted that these shock waves are smeared across tens or hundreds of cells, far more than the finite volume method presumably requires. Furthermore, the crest is reduced and the troughs are filled in: coarsening the mesh increases dissipation.

Applying the same experiment to the breaking dam scenario, as shown in Figure 2, yields an additional, unexpected twist.




Fig. 1. Performance of the Lax-Friedrichs method relative to n_{cells} for a Gaussian perturbation scenario with $t = 50s$, $\mu = 0.5$.




Fig. 3. Performance of the Lax-Friedrichs method relative to μ for a breaking dam scenario with $t = 50s$, $n_{cells} = 1000$.




Fig. 2. Performance of the Lax-Friedrichs method relative to n_{cells} for a breaking dam scenario with $t = 50s$, $\mu = 0.5$.

As the mesh becomes coarser, a nonphysical stairstep pattern emerges. This can be directly traced back to the update rule: Q_i^{n+1} depends on Q_{i+1}^n and Q_{i-1}^n , but not Q_i^n . As a result, the odd and even grid cells evolve independently from each other and any disparity between the two oscillates without dampening. Thus the stairstep pattern did not manifest in the Gaussian scenario because the initial condition was smooth.

The final experiment considers the effect of the timestep correction μ . It uses a fine mesh in order to reduce the dissipation described previously. As defined earlier, μ is a constant factor which scales down the timestep in order to maintain stability even in the presence of large nonlinearities. However, the behavior shown is remarkably counterintuitive: As the timestep is decreased, the solution becomes more dissipative, and consequently less accurate. When $\mu = 1.0$, the shock is crisp and almost vertical; when $\mu = 0.1$, the shock is effectively lost. The next section shall explore the cause of this phenomenon by showing how viscous behavior

can arise as a result of the numerical method used, rather than the underlying physical system.

VII. ARTIFICIAL VISCOSITY

The key idea behind artificial viscosity is that an approximate solution to a certain problem is oftentimes an exact solution to an approximation of that problem. The trick is to regard simple discretizations of PDEs as being building blocks which can be assembled into more sophisticated (and correspondingly more approximate) discretizations. One can then map the sophisticated discretization directly back to a modified PDE. In the case of Lax-Friedrichs, the approximation chosen for the flux function is effectively adding a diffusive term to the shallow-water equations.

Suppose that the ‘pure’ solution to the shallow water equations, $q_t + f(q)_x = 0$, corresponds to the naive flux function

$$F_{i+1/2} = \frac{1}{2} [f(Q_{i+1}) + f(Q_i)]$$

even though this is known to be unstable. Similarly, suppose that the ‘pure’ solution to the diffusion equation, $q_t = \beta q_{xx}$, corresponds to the flux function

$$F_{i+1/2} = -\beta \frac{Q_{i+1} - Q_i}{\Delta x}$$

This can be derived by a straightforward finite difference approximation. The Lax-Friedrichs flux function shown previously can be trivially rearranged as:

$$F_{i+1/2} = \frac{1}{2} [f(Q_i) + f(Q_{i+1})] - \frac{\Delta x}{2\Delta t} [Q_{i+1} - Q_i]$$

It is clear that this flux function pattern-matches the discretization of a different PDE:

$$q_t + f(q)_x = \frac{(\Delta x)^2}{2\Delta t} q_{xx}$$




Fig. 4. Comparison of height, momentum, and wavespeed over the spatial domain for a Gaussian perturbation with $t = 60s$, $\mu = 1.0$, $n_{cells} = 100$.

Thus the Lax-Friedrichs parameter $a = \frac{\Delta x}{\Delta t}$ controls the diffusion coefficient β according to

$$\beta = \frac{(\Delta x)^2}{2\Delta t} = \frac{1}{2}a\Delta x$$

This explains the two results described previously. Firstly, by taking a as constant, and reducing the cell size $\Delta x \rightarrow 0$, it is clear that $\beta \rightarrow 0$ and the viscous term in the PDE vanishes. Thus the Lax-Friedrichs discretization remains consistent with the pure shallow-water equations.

Secondly, reducing the timestep size Δt increases a , which then increases β . This explains the counterintuitive phenomenon where the solution becomes increasingly dissipative, and hence less accurate, as the timestep size is decreased. This then raises the question of whether it is possible to decouple β from Δt while still maintaining a consistent and stable discretization. The method described in the following section mostly achieves this.

VIII. THE LOCAL-LAX-FRIEDRICHS METHOD

Local-Lax-Friedrichs is a modification which retains the stability benefits of artificial viscosity while decreasing the resulting dissipation. In the basic Lax-Friedrichs, the viscosity parameter a is derived from the CFL condition via Δt . Although the timestep must be constant in space (local time stepping is not considered), the viscosity may vary. This yields advantages: in regions with shocks or other discontinuities, the presence of viscosity improves numerical stability, and in smooth regions, the lack of viscosity leads to a less dissipative and hence more accurate solution.

The viscosity parameter a is taken to be the maximum of the wavespeeds of either neighboring cell. As shown in Figure 4, this exhibits the desired behavior of being large in regions which are discontinuous or evolving quickly, and being small elsewhere.

$$a_{i-1/2} = \max(|f'(q)|) \quad \forall q \in (Q_{i-1}, Q_i)$$




Fig. 5. Performance of Local Lax-Friedrichs relative to μ for a breaking dam scenario with $t = 50s$, $n_{cells} = 1000$. Compare to Figure 3.

$$F_{i-1/2} = \frac{1}{2}[f(Q_{i-1}) + f(Q_i) - a_{i-1/2}(Q_i - Q_{i-1})] \quad (6)$$

From the CFL condition, it is clear that this approach is strictly less viscous than basic Lax-Friedrichs:

$$|f'(q)| \leq \frac{\Delta x}{\Delta t}$$

The Local Lax-Friedrichs method does not entirely break the coupling between dissipation and μ , but it strongly reduces it, as depicted in Figure 5. This experiment used the exact same parameters as Figure 3, substituting only the LxF kernel (Eq. 5 with the LLxF kernel (Eq. 6).

IX. COMPARISON OF LXF VS LLXF

A final round of experiments were conducted to compare Lax-Friedrichs to Local Lax Friedrichs. These indicate that Local Lax-Friedrichs is decidedly less dissipative and more accurate. Figure 6 shows the results of running LLxF with a middling $\mu = 0.4$ against LxF with different μ for the breaking dam scenario.

At $\mu = 0.4$, the LLxF method performs only slightly worse than LxF at $\mu = 1.0$. For any given choice of μ , the LLxF reaches a better solution in the same number of timesteps. The quality of the solutions converge as $\mu \rightarrow 1.0$. On the other hand, LLxF requires additional work to calculate $a(x)$ for each cell, though the amount is small and linear in the number of cells.

The staircase oscillations observed in the breaking dam case for LxF are not present for LLxf. This is presumably because the value of Q_i^n enters into the calculation of Q_i^{n+1} , albeit rather indirectly, via the max wavespeed λ_{max} and then the cell-wise flux $F_{i+1/2}^n$.




Fig. 6. Comparison of Lax-Friedrichs vs Local Lax-Friedrichs for a breaking dam scenario with $t = 50s$, $n_{cells} = 200$.




Fig. 7. Comparison of Lax-Friedrichs vs Local Lax-Friedrichs for a Gaussian perturbation scenario with $t = 200s$, $n_{cells} = 200$.

X. IMPLEMENTATION

This final section describes the design decisions behind the implementation. It is written in Julia, which permits the code to be both notably terse and closely mapped to the theory. The code organization parallels the theoretical development in the preceding sections.

The first design decision is the choice of data representation. Julia eschews standard object-oriented idioms such as inheritance and encapsulation in favor of a more functional and data-driven approach. Thus the state space Q is a simple mutable structure. The spatial domain $q(x)\forall x$ is defined as a one-dimensional array QS with inner type Q . The spatial domain has a ghost cell on either side to enforce boundary conditions, so the physical cell indices range over $[2, n_{cells} + 1]$.

These additional constraints could be expressed within the type system. Firstly, by using `OffsetArrays`, the indices can be cheaply translated to $[1, n_{cells}]$. Secondly, one may parameterize by the number of cells, $QS\{n_{cells}\}$, e.g. by

using `StaticArrays`, ensuring that the array size is known at JIT-compile time. This allows for array-size correctness checks and compiler optimizations. Finally, a custom constructor could encapsulate this and enforce the existence of the ghost cells. These features may prove helpful on larger projects, but were omitted here because they introduce more complexity than they resolve.

The choice of an array-of-structures representation is consistent with the theory, but disregards the fact that Q forms a vector space. This problem is readily resolved thanks to Julia’s *multiple dispatch*, where a function is resolved to a method by examining all of the types in the signature, rather than simply the ‘owning’ object. [4] This means that operations such addition can be extended to custom types in a unified way. Lines 12–26 extend addition and scalar multiplication over the state space type Q .

The general finite volume skeleton (Eq. 3) is given on lines 38–61. The skeleton accepts abstract strategies for flux computations, timestep computations, initial conditions, and boundary conditions. These strategies are implemented as first-class functions rather than as an object heirarchy. Lines 91–93 demonstrate a higher-order function which acts like a constructor for timestep strategies. Lines 100–102 demonstrate *currying* to convert a three-argument `cfl_dt()` into a two-argument version. Lines 104–112 demonstrate using the method’s *closure* to store arbitrary state, in this case the Δt calculated during the preceding function call. Finally, lines 147–148 ties everything together to run Lax-Friedrichs on a breaking dam scenario.

XI. CONCLUSION

The Lax-Friedrichs method is a simple finite volume technique for solving the shallow water equations. Starting from the fundamentals, this paper built up a reference implementation which leverages features of Julia to achieve terseness and clarity. An abstract finite volume skeleton allows different numerical methods to be swapped in. Two variants were implemented, Lax-Friedrichs and Local Lax-Friedrichs. The former suffers from excessive numerical viscosity, which the latter largely remedies. The causes and consequences of this viscosity were discussed with respect to two model problems, a Gaussian perturbation and a breaking dam. It is the author’s hope that this document may serve as a concise guide to students exploring this material in the future.

REFERENCES

- [1] O. Delestre, F. Darboux, F. James, C. Lucas, C. Laguerre, and S. Cordier, “FullSWOF: A free software package for the simulation of shallow water flows,” Mapmo, universit  d’Orl ans ; Institut National de la Recherche Agronomique, Research Report, Jan. 2014, 38 pages. [Online]. Available: <https://hal.archives-ouvertes.fr/hal-00932234>
- [2] R. LeVeque, *Finite Volume Methods for Hyperbolic Problems*, ser. Cambridge Texts in Applied Mathematics. Cambridge University Press, 2002. [Online]. Available: https://books.google.de/books?id=O_ZjpMSZiw0C
- [3] G. Strang, *Computational Science and Engineering*. Wellesley-Cambridge Press, 2007. [Online]. Available: <https://books.google.de/books?id=GQ9pQgAACAAJ>
- [4] J. Bezanson, A. Edelman, S. Karpinski, and V. B. Shah, “Julia: A fresh approach to numerical computing,” *SIAM Review*, vol. 59, no. 1, pp. 65–98, 2017. [Online]. Available: <https://doi.org/10.1137/141000671>

XII. APPENDIX: CODE LISTING

```

1  # Define precision
2  const Float = Float64
3
4  # Define state space datatype
5  mutable struct Q
6      h :: Float
7      hu :: Float
8  end
9
10 # Define spatial domain datatype
11 QS = Array{Q,1}
12
13 # Make Q a vector space
14 import Base.+, Base.-, Base.*
15
16 function +(lhs::Q, rhs::Q)
17     Q(lhs.h + rhs.h, lhs.hu + rhs.hu)
18 end
19
20 function -(lhs::Q, rhs::Q)
21     Q(lhs.h - rhs.h, lhs.hu - rhs.hu)
22 end
23
24 function *(a::Real, q::Q)
25     Q(a*q.h, a*q.hu)
26 end
27
28 "Shallow-water point-wise flux"
29 function f(q :: Q)
30     Q(q.hu, q.hu^2/q.h + 0.5*G*q.h^2)
31 end
32
33 "Shallow-water point-wise max wave speed"
34 function wavespeed(q :: Q)
35     u = q.hu / q.h
36     c = sqrt(G * q.h)
37     max(abs(u-c), abs(u+c))
38 end
39
40 "Finite volume skeleton"
41 function run(stoptime::Real, ncells::Int, choose_dt,
42             compute_fluxes!, apply_bcs!, apply_ics)
43
44     qs = apply_ics(ncells)
45     Fl = QS(ncells+2)
46     Fr = QS(ncells+2)
47
48     currenttime = 0
49     timesteps = 0
50     dx = Float(1000.0/ncells)
51
52     while currenttime < stoptime
53         apply_bcs!(qs)
54         dt = choose_dt(qs, dx)
55         compute_fluxes!(qs, Fl, Fr, ncells, dx, dt)
56         for x = 2:ncells+1
57             qs[x] -= dt/dx * (Fr[x] + Fl[x-1])
58         end
59         currenttime += dt
60         timesteps += 1
61     end
62     return (qs, currenttime, timesteps)
63 end
64
65 "Lax-Friedrichs kernel"
66 function lxf!(qs::QS, Fl::QS, Fr::QS,
67             ncells::Int, dx::Float, dt::Float)
68
69     a = dx/dt
70     for x = 2:ncells+2
71         q1,q2 = qs[x-1], qs[x]
72         f1,f2 = f(q1), f(q2)
73
74         Fr[x-1] = 0.5*((f2-f1) - a*(q2 - q1))
75         Fl[x-1] = 0.5*((f2-f1) + a*(q2 - q1))
76     end
77 end

```

```

75 "Local Lax-Friedrichs kernel"
76 function llxf!(qs::QS, Fl::QS, Fr::QS,
77             ncells::Int, dx::Float, dt::Float)
78
79     for x = 2:ncells+2
80         q1,q2 = qs[x-1], qs[x]
81         f1,f2 = f(q1), f(q2)
82
83         a = max(wavespeed(q1), wavespeed(q2))
84
85         Fr[x-1] = 0.5*((f2-f1) - a*(q2-q1))
86         Fl[x-1] = 0.5*((f2-f1) + a*(q2-q1))
87     end
88 end

```

```

89 # Timestepping strategies: (QS, dx) -> dt
90
91 function make_const_dt(dt::Float)
92     return (qs, dx) -> dt
93 end
94
95 function cfl_dt(qs::QS, dx::Real, mu::Real)
96     lambda = map(wavespeed,qs) |> maximum
97     return Float(dx / lambda * mu)
98 end
99
100 function make_cfl_dt(mu::Real)
101     return (qs, dx) -> cfl_dt(qs, dx, mu)
102 end
103
104 function make_lagging_dt(dt0::Real, mu::Real)
105     dt = dt0
106     function lagging_dt(qs, dx)
107         result = dt
108         dt = cfl_dt(qs, dx, mu)
109         return result
110     end
111     return lagging_dt
112 end

```

```

113 # Initial conditions: ncells -> QS
114
115 function gaussian_ics(ncells::Int)
116     qs = QS(ncells+2)
117     mid = ncells / 2.0 + 1
118     for x = 1:ncells+2
119         pos_rel = (mid-x) / (ncells/10.0)
120         qs[x] = Q(exp(-pos_rel^2) + 1, 0.0)
121     end
122     return qs
123 end
124
125 function breakingdam_ics(ncells::Int)
126     [x < ncells/2 ? Q(2,0) : Q(1,0) for x in 1:ncells+2]
127 end

```

```

128 # Boundary conditions: QS -> nothing
129
130 function outflow_bcs!(qs::QS)
131     qs[1] = qs[2]
132     qs[end] = qs[end-1]
133 end
134
135 function periodic_bcs!(qs::QS)
136     qs[1] = qs[end-1]
137     qs[end] = qs[2]
138 end
139
140 function reflecting_bcs!(qs::QS)
141     qs[1].h = qs[2].h
142     qs[1].hu = -qs[2].hu
143     qs[end].h = qs[end-1].h
144     qs[end].hu = -qs[end-1].hu
145 end

```

```

146 # Sample experiment run
147 qs,t,ts = run(50, 1000, make_cfl_dt(0.5),
148             lxf!, outflow_bcs!, breakingdam_ics)

```